// File: contracts/KompeteMarketplaceManager.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "@openzeppelin/contracts/access/AccessControl.sol";

import "./interfaces/IMarketplace.sol";

contract KompeteMarketplaceManager is AccessControl {
    enum ActionRequestType {
        // operator actions
        SetFeeRecipient,
        SetMintRecipient,
        AddCollection,
        RemoveCollection,
        SetLockDuration,
        // admin actions
        GrantAdmin,
        RevokeAdmin,
        GrantOperator,
        RevokeOperator,
        GrantPauser,
        RevokePauser,
        TransferOwnership
    }

    enum ActionRequestStatus {
        Submitted,
        Executed,
        Canceled
    }

    struct ActionRequest {
        ActionRequestType action;
        ActionRequestStatus status;
        uint256 date;
        uint256[] values;
        address[] addresses;
    }

    bytes32 public constant PAUSER_ROLE = keccak256("PAUSER_ROLE");
    bytes32 public constant OPERATOR_ROLE = keccak256("OPERATOR_ROLE");

    IMarketplace public immutable marketplace;

    uint256 public lockDuration;

    uint256 private _requestsCount;
    mapping(uint256 => ActionRequest) public requests;

    event ActionSubmitted(address indexed by, uint256 indexed requestId);
    event ActionExecuted(address indexed by, uint256 indexed requestId);
    event ActionCanceled(address indexed by, uint256 indexed requestId);

    event LockDurationChanged(address indexed by, uint256 duration);

    modifier onlyPausers() {
        require(hasRole(PAUSER_ROLE, _msgSender()), "Pauser role required");
        _;
    }

    constructor(IMarketplace marketplace_, address admin_) {
        require(address(marketplace_) != address(0), "Invalid marketplace address");
        require(admin_ != address(0), "Invalid admin address");

        marketplace = marketplace_;
        _setupRole(DEFAULT_ADMIN_ROLE, admin_);
        _setupRole(PAUSER_ROLE, admin_);
    }

    function _transferOwnership(address newOwner) private {
        require(newOwner != address(0), "Invalid address");
        marketplace.transferOwnership(newOwner);
    }

    function _setLockDuration(uint256 duration) private {
        lockDuration = duration;
        emit LockDurationChanged(msg.sender, duration);
    }

    function _setFeeRecipient(address recipient) private {
        require(recipient != address(0), "Invalid address");
        marketplace.setProtocolFeeRecipient(recipient);
    }

    function _setMintRecipient(address collection, address recipient) private {
        marketplace.setMintFeeRecipient(collection, recipient);
    }

    function _toggleCollection(address collection, bool allowed) private {
        marketplace.toggleCollection(collection, allowed);
    }

    function submit(
        ActionRequestType action,
        address[] calldata addresses,
        uint256[] calldata values
    ) external returns (uint256) {
        bool isAdmin = hasRole(DEFAULT_ADMIN_ROLE, _msgSender());
        if (isAdmin) {
            if (action == ActionRequestType.RevokeAdmin) {
                // prevent revoking self
                for (uint256 i = 0; i < addresses.length; i++) {
                    require(addresses[i] != msg.sender, "Can't revoke self admin");
                }
            }
        } else {
            require(action <= ActionRequestType.SetLockDuration, "Admin role required");
            require(hasRole(OPERATOR_ROLE, _msgSender()), "Admin or operator role required");
        }

        _requestsCount += 1;
        uint256 requestId = _requestsCount;

        requests[requestId] = ActionRequest({
            action: action,
            status: ActionRequestStatus.Submitted,
            date: block.number,
            values: values,
            addresses: addresses
        });

        emit ActionSubmitted(msg.sender, requestId);
        return requestId;
    }

    function execute(uint256 requestId) external {
        require(requests[requestId].status == ActionRequestStatus.Submitted, "Invalid Status");
        require(requests[requestId].date + lockDuration <= block.number, "Too soon");

        requests[requestId].status = ActionRequestStatus.Executed;
        ActionRequestType action = requests[requestId].action;

        if (action == ActionRequestType.SetFeeRecipient) {
            _setFeeRecipient(requests[requestId].addresses[0]);
        } else if (action == ActionRequestType.SetMintRecipient) {
            _setMintRecipient(requests[requestId].addresses[0], requests[requestId].addresses[1]);
        } else if (action == ActionRequestType.AddCollection) {
            _toggleCollection(requests[requestId].addresses[0], true);
        } else if (action == ActionRequestType.RemoveCollection) {
            _toggleCollection(requests[requestId].addresses[0], false);
        } else if (action == ActionRequestType.SetLockDuration) {
            _setLockDuration(requests[requestId].values[0]);
        }
        // admin
        else if (action == ActionRequestType.GrantAdmin) {
            for (uint256 i = 0; i < requests[requestId].addresses.length; i++) {
                _grantRole(DEFAULT_ADMIN_ROLE, requests[requestId].addresses[i]);
            }
        } else if (action == ActionRequestType.RevokeAdmin) {
            for (uint256 i = 0; i < requests[requestId].addresses.length; i++) {
                _revokeRole(DEFAULT_ADMIN_ROLE, requests[requestId].addresses[i]);
            }
        } else if (action == ActionRequestType.GrantOperator) {
            for (uint256 i = 0; i < requests[requestId].addresses.length; i++) {
                _grantRole(OPERATOR_ROLE, requests[requestId].addresses[i]);
            }
        } else if (action == ActionRequestType.RevokeOperator) {
            for (uint256 i = 0; i < requests[requestId].addresses.length; i++) {
                _revokeRole(OPERATOR_ROLE, requests[requestId].addresses[i]);
            }
        } else if (action == ActionRequestType.GrantPauser) {
            for (uint256 i = 0; i < requests[requestId].addresses.length; i++) {
                _grantRole(PAUSER_ROLE, requests[requestId].addresses[i]);
            }
        } else if (action == ActionRequestType.RevokePauser) {
            for (uint256 i = 0; i < requests[requestId].addresses.length; i++) {
                _revokeRole(PAUSER_ROLE, requests[requestId].addresses[i]);
            }
        } else if (action == ActionRequestType.TransferOwnership) {
            _transferOwnership(requests[requestId].addresses[0]);
        }
        // revert
        else {
            revert("Invalid action");
        }

        emit ActionExecuted(msg.sender, requestId);
    }

    function cancel(uint256 requestId) external {
        require(
            hasRole(DEFAULT_ADMIN_ROLE, _msgSender()) || hasRole(OPERATOR_ROLE, _msgSender()),
            "Admin or operator role required"
        );
        require(requests[requestId].status == ActionRequestStatus.Submitted, "Invalid Status");

        requests[requestId].status = ActionRequestStatus.Canceled;
        emit ActionCanceled(msg.sender, requestId);
    }

    function pauseMarketplace() external onlyPausers {
        marketplace.pause();
    }

    function unpauseMarketplace() external onlyPausers {
        marketplace.unpause();
    }

    function grantRole(bytes32 role, address account) public override {
        revert("Must use submit!");
    }

    function revokeRole(bytes32 role, address account) public override {
        revert("Must use submit!");
    }

    function renounceRole(bytes32 role, address account) public override {
        revert("Must use submit!");
    }
}

// File: @openzeppelin/contracts/access/AccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (access/AccessControl.sol)

pragma solidity ^0.8.0;

import "./IAccessControl.sol";
import "../utils/Context.sol";
import "../utils/Strings.sol";
import "../utils/introspection/ERC165.sol";

/**
 * @dev Contract module that allows children to implement role-based access
 * control mechanisms. This is a lightweight version that doesn't allow enumerating role
 * members except through off-chain means by accessing the contract event logs. Some
 * applications may benefit from on-chain enumerability, for those cases see
 * {AccessControlEnumerable}.
 *
 * Roles are referred to by their `bytes32` identifier. These should be exposed
 * in the external API and be unique. The best way to achieve this is by
 * using `public constant` hash digests:
 *
 * ```
 * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
 * ```
 *
 * Roles can be used to represent a set of permissions. To restrict access to a
 * function call, use {hasRole}:
 *
 * ```
 * function foo() public {
 *     require(hasRole(MY_ROLE, msg.sender));
 *     ...
 * }
 * ```
 *
 * Roles can be granted and revoked dynamically via the {grantRole} and
 * {revokeRole} functions. Each role has an associated admin role, and only
 * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
 *
 * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
 * that only accounts with this role will be able to grant or revoke other
 * roles. More complex role relationships can be created by using
 * {_setRoleAdmin}.
 *
 * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
 * grant and revoke this role. Extra precautions should be taken to secure
 * accounts that have been granted it.
 */
abstract contract AccessControl is Context, IAccessControl, ERC165 {
    struct RoleData {
        mapping(address => bool) members;
        bytes32 adminRole;
    }

    mapping(bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    /**
     * @dev Modifier that checks that an account has a specific role. Reverts
     * with a standardized message including the required role.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     *
     * _Available since v4.1._
     */
    modifier onlyRole(bytes32 role) {
        _checkRole(role);
        _;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IAccessControl).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) public view virtual override returns (bool) {
        return _roles[role].members[account];
    }

    /**
     * @dev Revert with a standard message if `_msgSender()` is missing `role`.
     * Overriding this function changes the behavior of the {onlyRole} modifier.
     *
     * Format of the revert message is described in {_checkRole}.
     *
     * _Available since v4.6._
     */
    function _checkRole(bytes32 role) internal view virtual {
        _checkRole(role, _msgSender());
    }

    /**
     * @dev Revert with a standard message if `account` is missing `role`.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     */
    function _checkRole(bytes32 role, address account) internal view virtual {
        if (!hasRole(role, account)) {
            revert(
                string(
                    abi.encodePacked(
                        "AccessControl: account ",
                        Strings.toHexString(uint160(account), 20),
                        " is missing role ",
                        Strings.toHexString(uint256(role), 32)
                    )
                )
            );
        }
    }

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) public view virtual override returns (bytes32) {
        return _roles[role].adminRole;
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _grantRole(role, account);
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _revokeRole(role, account);
    }

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been revoked `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) public virtual override {
        require(account == _msgSender(), "AccessControl: can only renounce roles for self");

        _revokeRole(role, account);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event. Note that unlike {grantRole}, this function doesn't perform any
     * checks on the calling account.
     *
     * [WARNING]
     * ====
     * This function should only be called from the constructor when setting
     * up the initial roles for the system.
     *
     * Using this function in any other way is effectively circumventing the admin
     * system imposed by {AccessControl}.
     * ====
     *
     * NOTE: This function is deprecated in favor of {_grantRole}.
     */
    function _setupRole(bytes32 role, address account) internal virtual {
        _grantRole(role, account);
    }

    /**
     * @dev Sets `adminRole` as ``role``'s admin role.
     *
     * Emits a {RoleAdminChanged} event.
     */
    function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
        bytes32 previousAdminRole = getRoleAdmin(role);
        _roles[role].adminRole = adminRole;
        emit RoleAdminChanged(role, previousAdminRole, adminRole);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * Internal function without access restriction.
     */
    function _grantRole(bytes32 role, address account) internal virtual {
        if (!hasRole(role, account)) {
            _roles[role].members[account] = true;
            emit RoleGranted(role, account, _msgSender());
        }
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * Internal function without access restriction.
     */
    function _revokeRole(bytes32 role, address account) internal virtual {
        if (hasRole(role, account)) {
            _roles[role].members[account] = false;
            emit RoleRevoked(role, account, _msgSender());
        }
    }
}

// File: contracts/interfaces/IMarketplace.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

interface IMarketplace {
    function toggleCollection(address collection, bool allowed) external;

    function setProtocolFeeRecipient(address recipient) external;

    function setMintFeeRecipient(address collection, address recipient) external;

    function pause() external;

    function unpause() external;

    function transferOwnership(address newOwner) external;
}

// File: @openzeppelin/contracts/access/IAccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/IAccessControl.sol)

pragma solidity ^0.8.0;

/**
 * @dev External interface of AccessControl declared to support ERC165 detection.
 */
interface IAccessControl {
    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {AccessControl-_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) external view returns (bool);

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {AccessControl-_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) external view returns (bytes32);

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) external;
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}